

OPENSTREETMAP : REQUÊTES OVERPASS

Jean-Christophe Becquet - **APITUX**

IUGA - 28/10/2022

RAPPEL :

DOCUMENTATION DES TAGS

- Les tags sont documentés
sur le [wiki OpenStreetMap](#)

RAPPEL :

INTERROGER LES OBJETS

- Icône ? sur le site openstreetmap.org

DÉCOUVRIR OVERPASS TURBO

Interroger la base de données OpenStreetMap

Overpass turbo

REQUÊTE GÉNÉRÉE AVEC L'ASSISTANT

```
[out:json][timeout:25];  
(  
node["amenity"="drinking_water"]({{bbox}});  
way["amenity"="drinking_water"]({{bbox}});  
relation["amenity"="drinking_water"]({{bbox}});  
);  
out body;  
>;  
out skel qt;
```

PREMIÈRE REQUÊTE ÉLÉMENTAIRE

```
node["amenity"="drinking_water"](bbox);  
out;
```

UNION : PARENTHÈSES

```
(  
node["amenity"="drinking_water"]({{bbox}});  
node["drinking_water"="yes"]({{bbox}});  
);  
out;
```

DÉFINIR UNE ZONE DE RECHERCHE

```
area["name"="Grenoble"]->.gre; // Attention ambigu  
// area["ref:INSEE"="38185"]->.gre;  
node["amenity"="drinking_water"](area.gre);  
out;
```


FILTRES COMBINÉS : SÉLECTIONNER SUR PLUSIEURS CRITÈRES

```
area["ref:INSEE"="38185"]->.gre;  
node["shop"]["wheelchair"="yes"](area.gre);  
out;
```

RÉCUPÉRER LA GÉOMÉTRIE DES WAY

```
area["name"="Parc Paul Mistral"]->.mistral;  
way["highway"="footway"](area.mistral);  
out geom;
```

REQUÊTE SUR DES SURFACES (WAY OU RELATION)

```
area["ref:INSEE"="38185"]->.gre;  
(  
way["leisure"="park"](area.gre);  
relation["leisure"="park"](area.gre);  
);  
out geom;
```

UTILISER UNE VARIABLE INTERMÉDIAIRE

```
area["ref:INSEE"="38185"]->.gre;  
node["amenity"="bank"](area.gre)->.bankd;  
node.bankd["atm"="yes"];  
out;
```

TROUVER LES ÉLÉMENTS MODIFIÉS ENTRE 2 DATES

```
area["ref:INSEE"="38185"]->.gre;  
node(changed:"2022-10-24T00:00:00Z", "2022-10-25T00:00:00Z")  
(area.gre);  
out;
```

TESTER L'ABSENCE D'UN TAG : !

```
area["ref:INSEE"="38185"]->.gre;  
node["shop"="bakery"][!"opening_hours"](area.gre);  
out geom;
```

COMPTER LES DONNÉES

```
area["ref:INSEE"="38185"]->.gre;  
(  
  node["man_made"="surveillance"](area.gre);  
  way["man_made"="surveillance"](area.gre);  
  relation["man_made"="surveillance"](area.gre);  
);  
out count;
```

TOUT RÉDUIRE À DES POINTS

```
area["ref:INSEE"="38185"]->.gre;  
(  
  node["amenity"="school"](area.gre);  
  way["amenity"="school"](area.gre);  
  relation["amenity"="school"](area.gre);  
);  
out center;
```


RÉCUPÉRER DES DONNÉES TABULÉES EN CSV

```
[out:csv (::id,::user,::timestamp,name,website,wheelchair)];  
area["ref:INSEE"="38185"]->.gre;  
(  
node["amenity"="cinema"](area.gre);  
way["amenity"="cinema"](area.gre);  
relation["amenity"="cinema"](area.gre);  
);  
out meta;
```

UN PEU PLUS COMPLEXE

```
area["ref:INSEE"="38185"]->.gre;  
node(area.gre)["highway"="bus_stop"]->.bus;  
node(around.bus:100)["amenity"="doctors"];  
//.bus out;  
out;
```

OVERPASS SOUS QGIS

Démonstration : faire des requêtes à partir de **QGIS**
avec l'extension **QuickOSM**

OVERPASS TURBO

Quelques ressources pour aller plus loin

- Carto'Cit  : [S rie de tutoriels OverpassAPI](#)
- Boris Mericskay : [Atelier prise en main de l'API Overpass](#)
- Julien Fastr  : [Conf rence des Geeks Anonyme](#)

TÉLÉCHARGER LES DONNÉES OPENSTREETMAP

Les exports de Geofabrik, licence ODbL

GeoDataMine

RAPPEL IMPORTANT CRÉDITER OPENSTREETMAP

Les contributeurs [OpenStreetMap](#), licence ODbL

Cette présentation a été réalisée
à partir du logiciel libre **Reveal.js**

Elle est partagée sous licence

Creative Commons BY-SA 4.0 International
(Attribution - Partage dans les Mêmes Conditions)



Jean-Christophe Becquet - **APITUX**